



Weld your way.

Carola - Exigences

Exigences pour le code du
programme Carola

Table des matières

1	Introduction	3
2	Efficacité globale des installations (OEE)	3
2.1	Déterminer la quantité réelle et la quantité de marchandises	3
2.1.1	Programme de master	3
2.1.2	Variables	6
2.1.3	Programme de composants simple	6
2.1.4	Gamme complexe de composants	8
3	Remarques	10
4	Variables	10
5	Termes	11

1 Introduction

Pour certaines fonctions de C-Gate, il est nécessaire que des informations soient fournies à C-Gate à partir du programme Carola. Veuillez vous référer aux chapitres correspondants pour les blocs et instructions nécessaires à la fonctionnalité en question.

Si vous utilisez la structure de données `ReadVar` ou la commande `OPCUAVAR` dans un autre contexte, veillez à ce que les domaines utilisés ne se chevauchent pas. Veuillez en outre noter que l'index utilisé dans le programme Carola n'est pas le même index sur l'interface web. Les valeurs sont transmises à C-Gate par la structure de données `ReadVar`. Dans l'automate, le premier index est un et le dernier 100. Dans C-Gate, le premier index est 0 et le dernier 99. Une valeur stockée sous l'index 50 dans l'automate peut être appelée sous l'index 49 dans C-Gate.

Veillez noter que les exemples de code Carola expliqués ci-dessous peuvent, le cas échéant, déjà être partiellement ou entièrement déposés dans le programme de master. Veuillez comparer les sections et les adapter si nécessaire. Un doublement des fonctions et des variables est reconnu comme une erreur par l'interpréteur Carola.

Remarque : les modifications des variables QIROX `WriteVar` peuvent influencer le déroulement du programme Carola.

Remarque : veuillez utiliser Adobe Acrobat Reader pour ouvrir ce fichier si vous souhaitez copier le code du programme Carola.

Remarque : veuillez n'utiliser que des chaînes de caractères d'une longueur maximale de 32 lettres. Dans le cas contraire, les valeurs ne seront traitées que partiellement, voire pas du tout.

2 Efficacité globale des installations (OEE)

2.1 Déterminer la quantité réelle et la quantité de marchandises

Les sections de programme suivantes doivent être stockées dans le programme maître et dans le programme de composant pour déterminer la quantité réelle et la quantité bonne. Il est important de noter que les procédures suivantes utilisent exclusivement la plage d'index 94 à 100 et ne peuvent pas être utilisées à d'autres fins. Cette plage peut encore s'étendre à l'avenir.

2.1.1 Programme de master

Les premières lignes sont nécessaires pour que les variables puissent être importées et écrites à partir d'autres programmes. Elles se situent juste après l'instruction `RESTART`.

```
RESTART
! DÉCLARATION DE LA VARIABLE
VAR SEL, RESULT
STRING VAR COMPSTAT, COMPQUAL
PUBLIC STRING VAR COMPTYPE, COMPVARI, COMPSERI, COMPSTEP, COMPSEAM, SEAMQUAL
```

Ce bloc définit une procédure qui transmet à C-Gate l'information qu'un composant commence. Au préalable, les variables utilisées ci-dessous doivent être définies dans le programme de composants.

```

PUBLIC PROC COMPSTART
COMPSTAT := ''
OPCUAVAR(0,96, COMPSTAT, 'COMPSTATUS')
WAITM(500)

STRHND(2, COMPTYPE, COMPTYPE, RESULT)
IF RESULT=0 THEN COMPTYPE:='NODATA'
STRHND (0,COMPTYPE, 'NODATA', RESULT)
IF RESULT=1 THEN BEGIN
LOPCUA1:
WRITE ('VARIABLE >COMPTYPE< PAS ENCORE DÉCRITE !')
WAITM(500)
PAUSE
JUMP LOPCUA1
END

COMPSTAT := 'START'
COMPQUAL := 'IN_PROGRESS'
COMPSEAM := ''
SEAMQUAL := ''
OPCUAVAR(0,100,COMPTYPE, 'COMPTYPE')
OPCUAVAR(0,99, COMPVARI, 'COMPVARIANT')
OPCUAVAR(0,98, COMPSERI, 'COMPSEARNO')
OPCUAVAR(0,97, COMPSTEP, 'COMPSTEPNAME')
OPCUAVAR(0,96, COMPSTAT, 'COMPSTATUS')
OPCUAVAR(0,95, COMPQUAL, 'COMPQUALITY')
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
OPCUAVAR(0,93, SEAMQUAL, 'COMPSEAMQUAL')
WAITM (300)
ENDP

```

Ce bloc définit une procédure qui transmet à C-Gate l'information qu'un composant est terminé.

```

! FIN DE L'ENREGISTREMENT DES COMPOSANTS
PUBLIC PROC COMPEND
COMPSEAM := ''
COMPSTAT := 'END'
OPCUAVAR(0,94, COMPSEAM, 'COMPSEAMNAME')
OPCUAVAR(0,96, COMPSTAT, 'COMPSTATUS')
WAITM(1000)

COMPTYPE := ''
COMPVARI := ''
COMPSERI := ''
COMPSTEP := ''
COMPSTAT := ''

```

```

COMPQUAL := ''
COMPSEAM := ''
SEAMQUAL := ''
OPCUAVAR(0,100,COMPTYPE,'COMPTYPE')
OPCUAVAR(0,99,COMPVARI,'COMPVARIANT')
OPCUAVAR(0,98,COMPSEMI,'COMPSEMI')
OPCUAVAR(0,97,COMPSTEP,'COMPSTEPNAME')
OPCUAVAR(0,96,COMPSTAT,'COMPSTATUS')
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
OPCUAVAR(0,94,COMPSEAM,'COMPSEAMNAME')
OPCUAVAR(0,93,SEAMQUAL,'COMPSEAMQUAL')
ENDP

```

Ce bloc définit une fonction qui transmet à C-Gate l'information qu'une nouvelle couture démarre. Le nom de la couture doit être écrit dans la variable `COMPSEAM` avant l'appel de la procédure.

```

! DÉBUT D'UNE SUTURE
PUBLIC PROC SEAMSTART
SEAMQUAL := ''
OPCUAVAR(0,94,COMPSEAM,'COMPSEAMNAME')
OPCUAVAR(0,93,SEAMQUAL,'COMPSEAMQUAL')
WAITM(100)
ENDP

```

Ce bloc définit une fonction qui transmet à C-Gate l'information que la couture en cours est terminée.

```

! FIN D'UNE COUTURE
PUBLIC PROC SEAMEND
COMPSEAM := ''
OPCUAVAR(0,94,COMPSEAM,'COMPSEAMNAME')
WAITM(100)
ENDP

```

Ce bloc définit une fonction qui demande sur le PHG si le composant est en ordre ou non. La vérification mise sur un contrôle visuel de l'ouvrier et bloque la suite du programme jusqu'à ce qu'une entrée soit effectuée. Vous pouvez adapter la formulation `EST-CE QUE LE COMPOSANT EST OK=1 OU PAS OK=0 ?` à vos besoins.

```

! QUALITÉ DE L'ENREGISTREMENT DES COMPOSANTS
PUBLIC PROC COMPQMAN
WREAD ('EST-CE QUE LE COMPOSANT EST OK=1 OU PAS OK=0 ?', SEL )
IF SEL=1 THEN COMPQUAL := 'OK' ELSE COMPQUAL := 'NOT_OK'
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
WAITM(100)
ENDP

```

Ce bloc définit une fonction qui définit automatiquement (sans interrogation) la qualité du composant sur en ordre. Cette fonction peut être utilisée si vous ne souhaitez pas saisir dynamiquement la qualité de l'efficacité globale de l'installation (OEE), mais que vous souhaitez tout de même que l'ensemble de la métrique soit calculé correctement.

```
! QUALITÉ DE L'ENREGISTREMENT DES COMPOSANTS
PUBLIC PROC COMPQAUT
COMPQUAL := 'OK'
OPCUAVAR(0,95,COMPQUAL,'COMPQUALITY')
WAITM(100)
ENDP
```

Ce bloc définit une fonction qui vérifie si la variable 'SEAMQUAL' est définie. Si elle n'est pas définie, la qualité est définie comme non OK. Si elle est activée, la qualité de la couture est fixée à la valeur correspondante. La variable 'SEAMQUAL' doit être écrite avant l'appel de la procédure.

```
! QUALITÉ DE LA COUTURE ACTUELLE. OK DOIT ÊTRE UTILISÉ POUR UNE BONNE QUALITÉ DE
  COUTURE ET NOT_OK POUR UNE MAUVAISE QUALITÉ.
PUBLIC PROC SEAMQAUT
STRHND(2,SEAMQUAL,SEAMQUAL,RESULT)
IF RESULT=0 THEN BEGIN
WRITE ('VARIABLE >SEAMQUAL< PAS ENCORE DÉCRITE !')
WAITM (500)
PAUSE
SEAMQUAL := 'NOT_OK'
END
OPCUAVAR(0,93,SEAMQUAL,'COMPSEAMQUAL')
WAITM (300)
ENDP
```

2.1.2 Variables

Les variables suivantes peuvent être utilisées dans le programme.

- **COMPTYPE** Type de composant.
- **COMPVARI** Variante de l'élément de construction.
- **COMPSEAM** : numéro de série du composant. Si aucune valeur n'est définie, C-Gate crée automatiquement un numéro de série à partir d'un horodatage.
- **COMPSTEP** : nom de l'étape de fabrication. Cette valeur est importante si le composant est usiné à plusieurs postes.
- **COMPSEAM** : nom de la couture. Une couture peut être composée de plusieurs segments de bande.
- **SEAMQUAL** Qualité de la couture. Peut être utilisé pour prendre en compte des résultats de mesure externes.

2.1.3 Programme de composants simple

Veuillez importer les variables et les procédures du programme master.

RESTART

```
! IMPORTATION DES VARIABLES ET DES PROCÉDURES
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSERI,COMPSTEP,COMPSEAM,SEAMQUAL FROM
MASTER
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTART,SEAMEND,SEAMQAUT FROM
MASTER
```

MAIN

Dans le déroulement du programme, activez au moins les champs obligatoires et, le cas échéant, également les champs facultatifs. Dès que tous les champs sont définis, appelez la procédure `COMPSTART`. Si vous avez acheté le module C-Gate.QM, vous pouvez en outre définir des coutures et structurer plus finement le protocole des composants. Utilisez pour cela les procédures `SEAMSTART` et `SEAMEND`. Si nécessaire, la qualité des coutures peut être définie manuellement à l'aide de la procédure `SEAMQAUT`. La procédure `SEAMEND` vous permet de terminer une couture. Dès que la pièce est terminée et que vous voulez définir la qualité de la pièce à partir du programme Carola, appelez soit `COMPQAUT` soit `COMPQMAN` pour transférer les données de qualité. Si vous avez acheté le module C-Gate.QM, la qualité de la pièce est automatiquement déterminée à l'aide de la surveillance QIROX SD. L'appel de la procédure `COMPQAUT` ou `COMPQMAN` n'a aucun effet sur le protocole des composants.

Pour terminer le composant, appelez la procédure `COMPEND`. L'ordre des appels est important.

Veuillez noter que jusqu'à une seconde doit s'écouler entre l'appel de `COMPEND` et le nouveau démarrage d'un composant avec `COMPSTART`, car le signal de fin peut sinon être perdu en raison d'une synchronisation des données dans la commande. Il est prévu que cette limitation soit supprimée dans l'une des prochaines versions du logiciel de la commande de robot.

```
! DÉFINITION DES INFORMATIONS SUR LES COMPOSANTS
! CHAMP OBLIGATOIRE
COMPTYPE := 'ARMOIRE ÉLECTRIQUE'

! EN OPTION
! COMPVARI := 'XL'

! EN OPTION
! COMPSERI := '1234567890'

! EN OPTION
! COMPSTEP := 'SOL'

! DÉMARRAGE DU COMPOSANT
CALL COMPSTART

! *****
! PROGRAMME DES ÉLÉMENTS DE CONSTRUCTION
! *****
```

```

COMPSEAM := 'SOUDURE D'ANGLE 1'
CALL SEAMSTART

! *****
! PROGRAMME SOUDURE D'ANGLE 1. PAR EXEMPLE, DIFFÉRENTS MOUVEMENTS DE GC.
! *****

! EN OPTION
! IF CONDITION THEN SEAMQUAL := 'OK' ELSE SEAMQUAL := 'NOT_OK'
! CALL SEAMQAUT

COMPSEAM := 'SOUDURE D'ANGLE 2'
CALL SEAMSTART

! *****
! PROGRAMME SOUDURE D'ANGLE 2. PAR EXEMPLE, DIFFÉRENTS MOUVEMENTS DE GC.
!
! EN OPTION
! SEAMQUAL := 'NOT_OK'
! CALL SEAMQAUT
! *****

CALL SEAMEND

! DÉFINIR LA QUALITÉ DE LA PIÈCE DANS LE PROGRAMME CAROLA
! APPEL AVANT COMPEND
CALL COMPQMAN
! OU
! CALL COMPQAUT

! FIN DE LA PIÈCE
! APPEL À COMPQMAN OU COMPQAUT
CALL COMPEND

! FIN DU PROGRAMME DES COMPOSANTS
END

```

2.1.4 Gamme complexe de composants

Le programme suivant Carola crée un composant avec deux étapes de fabrication. Les étapes de fabrication peuvent être soudées de manière séquentielle sur un robot ou en parallèle sur plusieurs. La condition préalable est que les étapes soient regroupées dans un protocole :

- Le type, la variante et le numéro de série doivent être définis de la même manière sur tous les robots.

- Le modèle de composant correspondant doit avoir le nombre minimum d'étapes de fabrication correct. Dans ce cas, deux étapes de fabrication sont soudées, c'est pourquoi le nombre doit être fixé au moins à la valeur deux.

```
RESTART
```

```
EXTERNAL STRING VAR COMPTYPE,COMPVARI,COMPSEAM,SEAMQUAL FROM  
MASTER
```

```
EXTERNAL PROC COMPSTART,COMPEND,COMPQMAN,COMPQAUT,SEAMSTART,SEAMEND,SEAMQAUT FROM  
MASTER
```

```
MAIN
```

```
! *****
```

```
! ÉTAPE DE FABRICATION 1
```

```
! *****
```

```
COMPTYPE := 'CHAUDIÈRE'
```

```
COMPVARI := '80KW'
```

```
COMPSEAM := '0042012342'
```

```
COMPSTEP := 'SOL'
```

```
CALL COMPSTART
```

```
COMPSEAM := 'SOUDURE D'ANGLE 1'
```

```
CALL SEAMSTART
```

```
! PROGRAMME SOUDURE D'ANGLE 1. PAR EXEMPLE, DIFFÉRENTS MOUVEMENTS DE GC.
```

```
COMPSEAM := 'SOUDURE D'ANGLE 2'
```

```
CALL SEAMSTART
```

```
! PROGRAMME SOUDURE D'ANGLE 2. PAR EXEMPLE, DIFFÉRENTS MOUVEMENTS DE GC.
```

```
CALL SEAMEND
```

```
CALL COMPQAUT
```

```
CALL COMPEND
```

```
! *****
```

```
! ÉTAPE DE FABRICATION 2
```

```
! *****
```

```
COMPTYPE := 'CHAUDIÈRE'
```

```
COMPVARI := '80KW'
```

```
COMPSEAM := '0042012342'
```

```
COMPSTEP := 'COUVERCLE'
```

```
CALL COMPSTART
```

```
COMPSEAM := 'SOUDURE D'ANGLE 3'
```

```
CALL SEAMSTART
```

```
! PROGRAMME SOUDURE D'ANGLE 3. PAR EXEMPLE, DIFFÉRENTS MOUVEMENTS DE GC.
```

```
COMPSEAM := 'SOUDURE D'ANGLE 4'
```

```
CALL SEAMSTART
```

```
! PROGRAMME SOUDURE D'ANGLE 4. PAR EXEMPLE, DIFFÉRENTS MOUVEMENTS DE GC.
```

```
CALL SEAMEND
```

```
CALL COMPQAUT
```

```
CALL COMPEND
```

```
END
```

3 Remarques

Si vous êtes ou avez été un client PDM ou UMS et que l'une des deux options est activée dans la configuration de la commande, il est impératif de tenir compte de la remarque suivante! Si l'une des deux options est activée mais que la commande du robot n'est pas connectée à l'application PC correspondante, des lacunes dans le flux de données apparaissent, qui peuvent durer jusqu'à 5 secondes et fausser significativement le protocole des composants. Veuillez soit connecter la commande du robot à l'application PC, soit faire supprimer l'option correspondante de la configuration.

Si vous travaillez avec plusieurs robots en même temps sur la même pièce et que vous souhaitez que toutes les données soient réunies dans un protocole de pièce dans le cadre du module C-Gate.QM, veuillez contacter le service CLOOS pour de plus amples informations. Dans le cas contraire, veuillez laisser le champ `COMPSEMI` vide, car cela pourrait entraîner des effets secondaires indésirables lors du calcul de l'efficacité globale de l'installation (OEE).

4 Variables

Vous avez la possibilité d'attacher des variables au protocole de composant à partir du programme Carola. Pour cela, il faut utiliser la commande `OPCUAVAR` (voir exemple). Si le nom (dernier paramètre) a le préfixe `'COMPVAR_'`, la variable est attachée au composant. Si le nom a le préfixe `'STEPVAR_'`, la variable est attachée à l'étape du composant. Si le nom a le préfixe `'SEAMVAR_'`, la variable est attachée à la couture actuelle. Pour chaque nom choisi, une seule valeur peut être stockée. La dernière valeur définie avec le nom en question est enregistrée dans le protocole. Le préfixe du nom n'est pas sauvegardé. La valeur doit être réinitialisée avant la fin de la couture, de l'étape ou de l'élément de construction. Si cela n'est pas fait, la valeur est également ajoutée à l'élément suivant. Il est possible de sauvegarder des nombres entiers, des nombres à virgule et des textes.

```
CALL SEAMSTART
```

```
OPCUAVAR(0,1, 'EXAMPLE-VALUE', 'COMPVAR_SEAM_MEASURE_A')
```

```
OPCUAVAR(0,1, '', '')
```

```
CALL SEAMEND
```

5 Termes

Un composant se compose de plusieurs éléments imbriqués et peut contenir des données de plusieurs robots. Les données des robots sont enregistrées dans des étapes de fabrication séparées. L'enregistrement d'une étape de fabrication est lancé par la procédure 'COMPSTART' et se termine par 'COMPEND'.

Une étape de fabrication est constituée de coutures. Une couture peut avoir un nom et est définie dans le programme de la pièce. Elle n'est enregistrée que lorsqu'une étape de fabrication est enregistrée. Elle peut être utilisée pour regrouper plusieurs segments logiquement liés. Une couture est démarrée et terminée par les procédures 'SEAMSTRT' et 'SEAMEND'.

Une couture est composée de segments. Un segment n'est enregistré que lorsque le robot enregistre une couture et qu'il exécute un mouvement GC ou que l'arc électrique est amorcé.

Un segment est composé de listes. Toute liste de soudure active dans le programme Carola pendant l'enregistrement d'un segment est également enregistrée. Ceci ne s'applique qu'aux listes basées sur QTI. Les données de soudure sont enregistrées et affichées de manière structurée par groupe ou par canal.